Rainy Day Activities for the Atari®

Nancy Mayer



Rainy Day Activities for the Atari®

Rainy Day Activities for the Atari®

Nancy Kozak Mayer, Ed.D
Electronic Playground and Widget Company, Inc.

A Reston Computer Group Book

Reston Publishing Company
A Prentice-Hall Company
Reston, Virginia

Library of Congress Cataloging in Publication Data

Mayer, Nancy Kozak.
Rainy day activities for the Atari.
1. Games — Data processing. 2. Atari computer — Programming. I. Title.
GV1469.2.M38 1983 794.8'2 83-9478
ISBN 0-8359-6390-X

Inside illustrations by Donna Ward

10 9 8 7 6 5 4 3 2 1

Printed in the United States of America

To Steve, who always believed I could do it

Contents

Acknowledgments

I wish to thank my husband, Steven T. Mayer, for believing that I had something to contribute to kids and computers long before I did and for his endless patience and boundless energy in giving me the skills to accomplish the task. Also, he's a fabulous bug zapper. I would also like to thank Judith Oksner who wrote the Introduction and whose imagination and wit has rescued me countless times, Bill Carris for so generously giving me the programs called "Hello!" and "Paddle Play," Tom Zimmerman for his gentle assistance as it got to be near crunch time, and Linda Kopp who cheerfully typed, and made the lay-outs and who has the nicest way of asking, "Are you sure you want to say THAT?"

To the Parents

This book is filled with fun and games on the ATARI® Home Computer for youngsters between three and nine years old. There are videogames, letter, word and number games, and graphic design and music activities.

The aim in writing this book was to make the shortest possible program listings so that parents or older children would not have to spend so much time typing in the program that there would be no time left over for having fun. For young children, it is a great effort to hunt and peck on the keyboard and try to keep their place in the program listing. Therefore we numbered the lines of the program 1, 2, 3 etc. instead of the conventional 10, 20, 30 etc. Since the zero is an unneccessary placeholder in these instances, it is often confused with the letter O and it is hard to find on the keyboard because it follows 9.

Some games are two-player; some are one player against himself or against the computer; some are free play where the only aim is the sheer pleasure of doing the activity.

Children's abilities sometimes vary widely in their early years, so there are suggestions for making most of the games harder or easier, if necessary. This book is not intended to teach programming, but looking at these changes can provide some information about basic game design.

Before You Begin

You're about to turn your ATARI® Home Computer into a toy store that will provide fifty games of music, art, letters, numbers, and nonsense for young children.

If you have never programmed a computer, don't worry! This book will be easy and fun. If you can read, you can put the Rainy Day Activities on your computer; if you can type that's even better! There's a difference between typing in a program and programming a computer. You'll be doing the former. We've already done the latter! And the programs are short, so they won't take long to enter. No listing has more than twenty lines.

A few of the games require no more than that your child be able to hold and move a joystick. Some demand a little more coordination, while still others require a familiarity with letters and numbers. And a few require your child to have reading skills — or to have an adult nearby! A suggested ability level is provided for some activities. But do feel free to experiment; some activities will be more challenging than others, and some will be just for fun.

All the games are flexible. They can be played with one or two players. You can raise or lower the difficulty level in keeping with your child's age and ability. All of these adjustments can be made by altering one or two lines of the program as you type it in. Instructions for these changes are included with each game.

How to Use This Book

Choose a game or activity and determine whether it seems suited to your child's age and skill level. Use a BASIC Language Cartridge in the cartridge slot.

Type the program listing *exactly* as it appears and hit the **RETURN** key at the end of each line. Remember, the computer is very fussy — if it sees ";" when it expects ":" you'll get an error message. If you get an ERROR message while typing the program, the white cursor should give a clue to the mistake. Carefully examine the book listing and the line you typed to find the problem, and then delete the line and retype it correctly.

Once you have RUN the program correctly, save it on disk or cassette to avoid having to retype it the next time your child is ready to play. Please refer to the ATARI® manual for instructions on saving programs on a cassette or a disk.

* * * * * *

Please read the HELP!! section that follows before you start to play. There are several Tricky Instructions you'll need to learn before typing in the programs

* * * * * *

Help!!

These notes are intended to give you some help in reading and entering the program listings. If you're interested in knowing what some of the shortcuts that were used, the most commonly used ones are below.

MOST COMMON PROBLEM: Nothing Happens!

Are you sitting there and nothing is happening???

1. Have you pressed the RETURN key? The RETURN key is the computer's commander. It shouts to the computer "Do this," "Remember this," "Change this," and so on. You must press RETURN at the end of each line each time you make a correction, and almost every time you use the keyboard to answer.

2. Plug in your joystick and then turn up your television volume if needed. The bottom front of the computer looks like this:

CONTROLLER JACKS



All of our games use jack 1 on the far left.

TRICKY INSTRUCTIONS

1. In order to make shortest possible listings we often had to use a trick involving the screen editor. The most common one is an instruction that tells the computer to "Clear the entire screen."

   In the program listing, the symbol looks like this: ↰

   The way to enter that instruction is tricky:

   a. Press the ESC key once; then . . .
   b. Hold down the CTRL key and at the same time press the CLEAR < key once.

   If you have done this correctly you will be left with the following symbol: ↰

2. We have also used the DELETE BACK S key, INSERT key, and up and down arrows, etc. in the program listings. This is so you can move across the screen without using advanced techniques like player/missile graphics.

   You type the listing similarly as described above. To get this symbol: ◁

   a. Press the ESC key once; then . . .
   b. Press the DELETE BACK S key once.

On the next page is a glossary of symbols used.

Glossary of Symbols

Symbol	*Keys*
	ESC CTRL CLEAR <
	ESC CTRL ↑ -
	ESC CTRL ↓ =
	ESC CTRL → *
	ESC CTRL ← +
	ESC SHIFT INSERT >
▶	ESC CTRL INSERT >

Note: When graphics symbols are called for, you don't need to press ESC first. For example, hold down CTRL and press [, and you'll end up with a heart like this: ♥

Programming Conventions

Many of the programs are densely packed but if you are learning programming and want to try to wade through the programs, here are some conventions we tried to use when it was convenient.

- ? is shorthand instead of writing out PRINT
- Graphics 23 (or some other big number)
  Adding 16 to a graphics mode gets rid of the text window. So this is really graphics mode 7 plus 16 equals Gr. 23.
- ?#6; This allows you to "write" above the text window area in graphics modes 1 and 2.
- Poke 752,1 This gets rid of the cursor.
- Poke 764,255 This clears out the last key pressed. That way, if a person mistakenly hits two keys instead of one, the second one will be cleared instead of held over where it will interfere with the next question.
- Poke 18,0 This sets the computer's internal clock to zero so that we can have a
  Poke 19,0 timer that starts when the game starts.

Some Abbreviations Used

T	time
S	score (sometimes sound)
H	horizontal
V	vertical
R	random
L	letter
W	word
A	answer
C	count
N	number (sometimes name)

If you still have a question about how these programs work, write to us at:

The Electronic Playground and Widget Company, Inc.
24 Main Street
Essex, Connecticut 06426

Rainy Day Activities for the Atari®

Games

100 Years Old

Do you know when you'll be 100 years old? How about your parents? The computer will tell you . . .

Just tell the computer what your name is and how old you are. Then it will tell you when you'll be 100 years old!

Age Program

```
1  ? " ↰ HOW OLD ARE YOU?
2  INPUT A
3  Y=1983+(100-A)
4  ? "IN  ";Y;" YOU WILL BE 100 YEARS OLD!
```

Variations

Make sure you change line 3 when the year changes from 1983 to 1984, etc.

Old line 3: Y = 1983 + (100 − A)

New line 3: Y = 1984 + (100 − A)

How It Works

1. Type in the program or load from disk or cassette.
2. Type RUN.

How It Looks



3. Type in your age.



4. The computer tells you the year when you will be 100 years old.



5. To begin again, type RUN.
6. Save on disk or cassette as *Age*.

Popcorn

Popcorn is popping out of the pan. You can catch it with your scoop. How much popcorn can you catch before it stops popping?

You will move the joystick from side to side to scoop up the popcorn. You will get one point for each popcorn kernel you catch.

Popcorn Program

```
1  OPEN #2,12,0,"S:":POKE 752,1
2  ? " ↰ MOVE JOYSTICK TO CATCH THE POPCORN
3  ? "PRESS RED BUTTON TO START
4  IF STRIG(0)=1 THEN 4
5  ? " ↰ ":H=20:S=0:T=200
6  POS. 10,0:? "SCORE..";S;"       TIME..";T;" "
7  R=180*RND(0)+2
8  IF R<33 THEN POS. R,2:? "@"
9  POS. 0,2:? " ↓ " Note: To make this press ESC and hold down SHIFT INSERT.
10  FOR C=1 TO 4
11  IF STICK(0)=7 THEN H=H+1:IF H>32 THEN H=32
12  IF STICK(0)=11 THEN H=H-1:IF H<1 THEN H=1
13  POS. H,23:? " W "; Note: To make this hold down CTRL and press GFGF.
14  POS. H+C,22:GET #2,Z
15  IF Z < >32 THEN S=S+1
16  IF Z=32 THEN SO. 0,253,6,14:SO. 0,0,0,0
17  NEXT C
18  T=T-1
19  IF T>=0 THEN 6
20  POS. 16,2:? "GAME OVER
```

Variations

To make the game easier, list the program and change line 7 to make more popcorn fall.

Old line 7: R = 180*RND(0) + 2
New Line: R = 96*RND(0) + 2

How It Works

1. Type in program or load from disk or cassette.
2. Plug joystick into jack 1.
3. Turn on TV volume.
4. Type RUN.
5. Press red button.

How It Looks



6. Move the joystick from side to side to catch the popcorn.



7. The game is over when the time runs out.



8. To start over, type RUN.
9. Save on disk or cassette as *Popcorn*.

Tally Sheet

How many did you catch?

Name	*Day*	*Score*	*Name*	*Day*	*Score*



Know-It-All

You can ask your most private questions. Will you pass your test? Does your boyfriend like you? Are you super at video games? The Know-It-All will search down into its knowledge and tell you its truest answer.

Knowall Program

```
1 DIM Q$(100)
2 ? " ↰ TYPE IN YOUR QUESTION.
3 ? "I, THE GREAT KNOW-IT-ALL, WILL TRY TO ANSWER.
4 INPUT Q$
5 N=INT(6*RND(0)+1)
6 IF N=1 THEN ? "YES! ABSOLUTELY!
7 IF N=2 THEN ? "NO! NO CHANCE!
8 IF N=3 THEN ? "PERHAPS . . . I CANNOT BE SURE.
9 IF N=4 THEN ? "MAYBE YES . . . MAYBE NO.
10 IF N=5 THEN ? "YES! IT SEEMS LIKELY.
11 IF N=6 THEN ? "NO! I DON'T THINK SO.
12 ? :GOTO 4
```

Variations

You can change the answers by changing lines 6 through 11.

Old line 6: IF N = 1 THEN ? "YES! ABSOLUTELY!

New line 6: IF N = 1 THEN ? "OF COURSE, TURKEY!

How It Works

1. Type in program or load it from disk or cassette.
2. Type RUN.
3. Type in your question.
4. Press **RETURN** .

How It Looks

```
TYPE IN YOUR QUESTION.
I, THE GREAT KNOW-IT-ALL,
WILL TRY TO ANSWER.
? AM I THE FAIREST IN THE
LAND?
```

5. You'll get an immediate answer and you can then ask another question.

```
TYPE IN YOUR QUESTION.
I, THE GREAT KNOW-IT-ALL,
WILL TRY TO ANSWER.
? AM I THE FAIREST IN THE
LAND?
YES! ABSOLUTELY!
```

6. To end, press **SYSTEM RESET** .
7. Save on disk or cassette as *Knowall.*

Lasso The Stars

Nine stars will sparkle on the screen. How quickly can you lasso them?

Move the joystick so that the cursor is over the star. Press the red button to lasso it. The bell means "Atta, Cowpoke!" The razzberry sound means "Keep on trying!" The score is how long it took you. Low score wins.

Lasso Program

```
1 OPEN #2,12,0,"S:":SET. 2,0,0
2 ? "LASSO THE STARS
3 ? "PUSH RED BUTTON TO PICK UP
4 FOR T=1 TO 1200:NEXT T:? " ↰ "
5 FOR C=1 TO 9
6 H=30*RND(0)+2:V=2*C
7 POS. H,V:? "*";:NEXT C:N=0:R=0
8 FOR T=1 TO 50:NEXT T:J=STICK(0):R=R+1
9 IF J=11 THEN ? "←";
10 IF J=7 THEN ? "→";
11 IF J=14 THEN ? "↑";
12 IF J=13 THEN ? "↓";
13 IF STRIG(0)=1 THEN 8
14 GET #2,A
15 IF A< >170 THEN S=6:GOTO 17
16 S=10:N=N+1:? " ◁ ";
```



Note: To type this character, press ESC once and then press DELETE BACK S.

```
17 SO. 0,150,S,8
18 FOR T=1 TO 100:NEXT T
19 SO. 0,0,0,0:IF N< >9 THEN 8
20 ? "  SCORE=";INT(R/10)
```

How It Works

1. Type in the program or load from disk or cassette.
2. Plug joystick in jack 1.
3. Turn up TV volume.
4. Type RUN.
5. Move joystick so that your magic square is over star. Press red button to pick star up. Bell means "right." Buzzer means "try again."

How It Looks



6. Score is the time it took you. Low score wins.



7. To play again, type RUN .
8. Save on disk or cassette as *Lasso*.

Tally Sheet

Name	Day	Score	Name	Day	Score

Heads or Tails

This is a guessing game! The computer flips a coin. You have to guess if it flipped *heads* or *tails*. Can you get all ten guesses right?

All you have to do is type H if you think it will flip *heads* or T if you think it will flip *tails*. The game is over after ten rounds.

Coin Program

```
1 DIM A$(1)
2 S=0
3 FOR C=1 TO 10
4 ? " ↰ ROUND  ";C
5 ? "I'M GOING TO FLIP A COIN.
6 ? "YOU GUESS HEADS OR TAILS.
7 POKE 764, 255
8 INPUT A$
9 N=INT(2*RND(0))
10 IF N=0 AND A$="H" THEN 13
11 IF N=1 AND A$="T" THEN 13
12 ? "NOPE, SORRY.":GOTO 14
13 ? "RIGHT YOU ARE!":S=S+1
14 ? "                    SCORE  ";S
15 FOR T=1 TO 1800
16 NEXT T
17 NEXT C
18 ? "GAME IS OVER
```

Variations

To make the game move more quickly, change line 15 so that there will be less time between rounds.

Old line 15: FOR T = 1 TO 1800

New line 15: FOR T = 1 TO 900

How It Works

1. Type in program or load it from disk or cassette.
2. Type RUN.
3. Type in H if you think it will be HEADS. Type in T if you think it will be TAILS.
4. Press RETURN .

How It Looks



5. You find out if you're right or not. How well did you guess?



6. After a pause, it is the next round.
7. The game ends after Round 10.
8. To begin again, type RUN.
9. Save on disk or cassette as *Coin*.

Tally Sheet

What's your best score?

Name	Day	Score	Name	Day	Score

Lift-Off

Time is important in outer-space trips. In this game the timer in the space control center is broken. You must guess time correctly for the lift-off.

The computer will tell you how many seconds till lift-off. You must guess the number of seconds that have passed. Press the SPACE BAR to stop the counting. The computer will tell you how close you were and it will give you your score. The best score is 100.

Lift-Off Program



```
1  OPEN #2,4,0,"K:"
2  ? " ↰ PRESS START"
3  IF PEEK(53279)=7 THEN 3
4  T=INT(10*RND(0)+1)
5  ? " ↰ ":? :?
6  ? T;"  SECONDS"
7  SET. 2,T,2
8  ? "!!!!!!!!!! NOW TIMING  !!!!!!!!!!
9  ? :?: ?: ? "PRESS SPACE BAR TO STOP
10  POKE 19,0:POKE 20,0
11  GET #2,K
12  IF K< >32 THEN 10
13  A=(1/60)*PEEK(20)+(256/60)*PEEK(19)
14  A=INT(A)
15  ? "YOUR TIME WAS . . . .";A;"  SEC.
16  ? "YOUR SCORE IS . . . .";INT(100-ABS((A-T)/T)*100)
17  ?  :?  :? "PRESS RETURN
18  GET #2,K:IF K< >155 THEN 17
19  GOTO 2
```

How It Works

1. Type in program or load it from disk or cassette.
2. Type RUN.
3. Press the **START** key. The computer tells you the number of seconds to guess. Guess when 8 seconds are up!

How It Looks



4. Press the space bar when you think 8 seconds have passed.



In this example, the player pressed the space bar after 4 seconds passed. His score was 50%. Can you score 100%?



5. To begin again, press the RETURN key.
6. To end, press **SYSTEM RESET**
7. Save on disk or cassette as *Lift-Off*.

Tally Sheet

How many times did you get 100%?

Name	Day	Score	Name	Day	Score

Ski

Ski around trees and moguls without hitting anything. The track gets narrower and narrower and narrower . . .

Each time you ski, a new trail appears. You use the joystick to move through the path without hitting anything. When you hit, there is a crashing sound. If you crash, your turn is over. The trail gets narrower with each round. How difficult a trail can you ski?

Ski Program

```
1  OPEN #2,12,0,"S:":SET. 2,0,14:FOR R=1 TO 10
2  ? " ↰ ROUND  ";R:V=10
3  FOR H=35 TO 5 STEP -1:POKE 752,1
4  V=V+(INT(6*RND(0)-2))
5  IF V>23 THEN V=23
6  IF V<12 THEN V=12
7  POS. H,V:? "*";
8  POS. H,V-(10-R):? "*";:NEXT H
9  H=0:V=V-4
10  J=STICK(0)
11  IF J=11 THEN H=H-1
12  IF J=7 THEN H=H+1
13  IF J=14 THEN V=V-1
14  IF J=13 THEN V=V+1
15  H=H+1:POS. H,V
16  GET #2,A:IF A=42 THEN SO. 1,253,6,14
17  POS. H,V:? ">";:IF H>35 THEN NEXT R
18  FOR T=1 TO 90:NEXT T
19  SO. 1,0,0,0:IF A=42 THEN END
20  GOTO 10
```

Variations

To make the skier move more slowly change line 18.

Old line 18: FOR T = 1 TO 90:NEXT T

New line 18: FOR T = 1 TO 180:NEXT T

How It Works

1. Type in the program or load it from disk or cassette.
2. Plug joystick into jack 1.
3. Turn up TV volume.
4. Type RUN.

How It Looks



5. You must ski in the trail between the snowflakes. Move the joystick to avoid hitting the flakes. When you hit, there is a crashing sound. If you crash, your turn is over.



6. How difficult a trail can you ski?
7. To start again, hit **SYSTEM RESET**
8. Save on disk or cassette as *Ski.*

Tally Sheet

How difficult a trail can you ski?

Name	Day	Score	Name	Day	Score

Copycat

This one is a joke! Type whatever you want and see what a copycat the computer is! There is only one way to trip up the computer. Do you know how?

Copycat Program

```
1 DIM A$(100)
2 ? " ↰ HI! I'M A COPYCAT.
3 ? "TYPE SOMETHING IN. I'LL SHOW YOU.
4 INPUT A$
5 ? " ";A$;" "
6 GOTO 4
```



How It Works

1. Type in program or load from disk or cassette.
2. Type RUN.
3. Type in whatever you want.
4. Press **RETURN** .
5. You can continue typing in things to be copied for as long as you want. Just remember to press **RETURN** at the end of each thing you want copied.
6. Press **SYSTEM RESET** to end.
7. To begin again, type RUN.
8. Save on disk or cassette as *Copycat*.

How It Looks

```
HI! I'M A COPYCAT.
TYPE SOMETHING IN. I'LL
SHOW YOU.
? ■
```

```
HI! I'M A COPYCAT.
TYPE SOMETHING IN. I'LL
SHOW YOU.
? AREN'T YOU SILLY!
```

```
HI! I'M A COPYCAT.
TYPE SOMETHING IN. I'LL
SHOW YOU.
AREN'T YOU SILLY.
AREN'T YOU SILLY.
? ■
```

Shapes Up

Two complicated shapes appear. You decide if they are the same or different. Watch the designs being drawn. Type "S" if they are the same. Type "D" if they are different. There are ten pairs in a game. How many of the ten can you get right?

Shapes Program

```
1  DEG :S=0:DIM A$(1):FOR C=1 TO 10:GR. 7
2  P=INT(RND)(0)*2):N=INT(RND(0)*14+12)
3  H=40:V=38:GOSUB 17
4  N=N-3:V=35:GOSUB 17
5  N=N-6:V=34:GOSUB 17
6  N=N+9
7  IF P< >1 THEN N=INT(RND(0)*14+12)
8  H=100:V=38:GOSUB 17
9  N=N-3:V=35:GOSUB 17
10  N=N-6:V=34:GOSUB 17
11  ? C;:? " SAME OR DIFFERENT?":POKE 764,255:INPUT A$
12  IF A$="S" AND P=1 THEN ? "RIGHT!":S=S+1:GOTO 15
13  IF A$="D" AND P< >1 THEN ? "RIGHT!":S=S+1:GOTO 15
14  ? "NOPE!"
15  ? "SCORE=";S;:FOR T=1 TO 900:NEXT T
16  NEXT C:? "GAME OVER":END
17  FOR D=0 TO 360 STEP 360/N:COLOR 1
18  PLOT SIN(D)*20+H,COS(D)*15+V
19  DRAWTO SIN(360/N+D)*20+H,COS(360/N+D)*15+V
20  NEXT D:RETURN
```

Variations

To shorten the game, change line 1.

Old line 1: DEG:S = 0:DIM A$(1):FOR C = 1 TO 10:GR. 7

New line 1: DEG:S = 0:DIM A$(1):FOR C = 1 TO 5:GR. 7

How It Works

1. Type in the program or load from cassette or disk.
2. Type RUN.
3. Watch the designs being drawn.
4. Type in S for Same and D for Different.
5. Press **RETURN**

How It Looks



6. The computer tells you if you're right or not and gives your score.



7. Game ends in ten rounds.
8. To play again, type RUN.
9. To end, hit **SYSTEM RESET**
10. Save on disk or cassette as *Shapes*.



Tally Sheet

What is your best score?

Name	Day	Score	Name	Day	Score

Four Leaf Clover

I'm looking over a four leaf clover
That I've overlooked before . . . .

There is a four leaf clover hiding somewhere on the screen. How many steps does it take you to find it? Move the joystick around the screen till you discover the four leaf clover. When it appears, you will be told how many steps it took you. Try to do it in as few steps as possible.

Clover Program

```
1 A=B=S=0:GR. 0
2 ? "FIND THE FOUR LEAF CLOVER
3 ? "PRESS RED BUTTON TO START
4 IF STRIG(0)=1 THEN 4
5 GR. 0:SET. 2,12,4
6 H=35*RND(0):V=15*RND(1)
7 POS. 20,10:? " ";
8 J=STICK(0)
9 IF J=14 THEN ? "↑↑";:S=S+1
10 IF J=13 THEN ? "↓↓";:S=S+1
11 IF J=7 THEN ? "→";:S=S+1
12 IF J=11 THEN ? "← ← ←";:S=S+1
13 IF J=15 THEN 8
14 X=PEEK(85):Y=PEEK(84)
15 IF X>H AND X<H+3 THEN A=1
16 IF Y>V AND Y<V+6 THEN B=1
17 IF A*B=1 THEN 20
18 ? " . ";
19 GO TO 8
20 ? " ♣ ";S
```



Note: Hit ESC once, hold down CTRL and press "P" once to make clover.

Variations

If you want to make the game easier, hit SYSTEM RESET . List the program. Change lines 15 and 16. These lines describe an area of the screen within whose borders the clover lies. Making the numbers bigger makes the area bigger, which makes the game easier. Save the new program.

Old line 15: IF X > H AND X < H + 3 THEN A = 1

New line 15: IF X > H AND X < H + 10 THEN A = 1

Old line 16: IF Y > V AND Y < V + 6 THEN B = 1

New line 16: IF Y > V AND Y < V + 18 THEN B = 1

How It Works

1. Type in the program or load it from the disk or cassette.
2. Plug joystick in jack 1.
3. Type RUN.
4. Press red button on joystick.

How It Looks



5. Move joystick to find clover.
6. This player found the clover in seven steps.



7. Type RUN *to* play again.



8. To end, press SYSTEM RESET
9. Save on disk or cassette as *Clover*.

Tally Sheet

What's your lowest score?

Name	Day	Score	Name	Day	Score

Snowfall Ending

The snowfall is ending and the snowflakes are getting harder to catch. Move the joystick from side to side to catch the snowflakes before they reach the ground. At first, three snowflakes come down at once. After you catch ten of them, they start to come down two at a time. When your score is 20, they come down one at a time. How many can you catch before your time is up?

Snow Program

```
1  OPEN #2,12,0,"S:":POKE 752,1:H=20:S=0:T=1000
2  ? "  MOVE YOUR JOYSTICK TO CATCH THE SNOWFLAKES
3  ? "PRESS RED BUTTON TO START
4  IF STRIG(0)=1 THEN 4
5  ?  "  "
6  POS. 10,0:? "SCORE..";S;"     TIME..";T;"  "
7  C=500*RND(0)+3
8  POS. 0,2:? "■":IF C>35 THEN 12
9  POS. C,2:IF S>19 THEN ? "*"
10  IF S<10 THEN ? "***": GOTO 12
11  IF S<20 THEN ? "**"
12  IF STICK(0)=11 THEN H=H-1
13  IF STICK(0)=7 THEN H=H+1:IF H>36 THEN H=36
14  IF H<2 THEN H=2
15  POS. H,22:? "♦"
    Note: To make this character hold down CTRL and press [ . ] key.
16  POS. H+1,21:GET #2,Z:POS. H+1,21:PUT #2,Z:IF Z< >32 THEN S=S+1
17  T=T-1
18  IF T>0 THEN 6
19  POS. 21,0:? "GAME OVER":END
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Plug the joystick into jack 1.
3. Type RUN.
4. To begin, press the red button on the joystick.

How It Looks



5. Move the joystick from side to side. Only direct hits are counted in the score.



6. The game ends when time runs out.



7. To begin again, type RUN.
8. Save on disk or cassette as *Snow*.

Tally Sheet

What's your best score?

Name	Day	Score	Name	Day	Score

Letter Games

Letter Shoot

Two letters are on the screen. Press the red button to shoot if they are the same. Move the joystick up to get rid of the letters if they are not the same. If you are right, you will hear the shot. If you are wrong, you will hear the shot miss. There is a counter to tell you how many you have right.

For kids who are learning to match letters. This is the easiest letter game.

Shoot Program

```
1  ? " ↰ PRESS RED BUTTON TO SHOOT LETTERS THAT ARE THE
   SAME.
2  ? "MOVE THE JOYSTICK UP TO GET RID OF        LETTERS THAT
   ARE DIFFERENT.
3  ? "PRESS RED BUTTON TO START . . GOOD LUCK":S=0
4  IF STRIG(0)=1 THEN 4
5  GR. 18
6  POKE 18,0:POKE 19,0
7  SO. 1,0,0,0
8  IF PEEK(19)>10 THEN 16
9  L1=21*RND(0)+65:L2=21*RND(0)+65
10  P=RND(0):IF P>0.3 THEN L2=L1
11  POS. 9,5:? #6;CHR$(L1);"  ";CHR$(L2)
12  IF STICK(0)=14 THEN 9
13  IF STRIG(0)=1 THEN 12
14  IF L1=L2 THEN FOR T=1 TO 100:SO. 1,T,6,10:NEXT T:S=S+1:POS. 9,9:?
    #6;S: GOTO 7
15  FOR T=1 TO 100:SO. 1,60,6,10:NEXT T:GOTO 7
16  POS. 5,5:? #6;"GAME OVER":GOTO 16
```

How It Works

1. Type in the program or load from disk or cassette.
2. Plug joystick in jack 1.
3. Turn TV volume up.
4. Press red button on the joystick when you're ready.
5. Press red button.
6. Move joystick up.
7. Press red button.
8. To start over, hit the SYSTEM RESET key.
9. Save on disk or cassette as *Shoot*.

How It Looks









Tally Sheet

What's your best score?

Name	Day	Score	Name	Day	Score

Let's Draw

Have you ever drawn a picture with letters? If you match the *small* letter with its *BIG* letter in this game, you can draw with them.

You'll see a small letter on the screen. Type in its BIG letter. If you're wrong, you'll hear a razzberry and another small letter will flash on the screen.

If you type in the right letter, you can draw a picture by moving the joystick all around. The computer will flash to the next letter when your drawing time is up.

For kids who recognize letters of the alphabet.

Caps Program

```
1  DIM U$(1),L$(1),A$(1)
2  POKE 752,1:SET. 2,0,0
3  L=INT(25*RND(0))+97:L$=CHR$(L)
4  U=L-32:U$=CHR$(U)
5  ? " ↰ ":POS. 10,10:? L$
6  POS. 10,20
7  ? "TYPE IN THE BIG LETTER.
8  POS. 10,21:INPUT A$
9  POS. 13,10:? A$:FOR T=1 TO 200:NEXT T
10  IF U$< >A$ THEN FOR T=1 TO 60:SO. 1,60,8,10:NEXT T:SO.
    1,0,0,0:GOTO 8
11  IF U$=A$ THEN FOR T=1 TO 120:SO. 1,30,10,10:NEXT T:SO. 1,0,0,0
12  ? " ↰ USE THE JOYSTICK TO DRAW
13  POKE 18,0:POKE 19,0
14  J=STICK(0)
15  IF J=11 THEN ? "←←";U$;L$;"←←";
16  IF J=7 THEN ? U$;L$;
17  IF J=14 THEN ? "↑↑";U$;L$;"↑↑";
18  IF J=13 THEN ? "↓↓";U$;L$;"↓↓";
19  IF PEEK(19)>5 THEN 3
20  GOTO 14
```

Variations

To shorten drawing time, decrease the time in line 19.

Old line 19: IF PEEK(19) > 5 THEN 3

New line 19: IF PEEK(19) > 2 THEN 3

To lengthen drawing time, increase 5 to 10 in line 19.

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Turn TV volume up.
4. Type in the capital (BIG) letter.
5. Press **RETURN** .

How It Looks



6. If you are right, you hear a bell and can draw with the joystick. If you are wrong, you hear a razzberry and can keep trying.



7. Move the joystick up, down, and all around.



8. When time is up, a new letter flashes on.
9. To end, press the **SYSTEM RESET** key.
10. Save on disk or cassette as *Caps*.

8. When time is up, a new letter flashes on.



9. To end, press the SYSTEM RESET key.

10. Save on disk or cassette as *Caps*.

Missing

In the middle of the night a terrible thief came along and stole letters from the alphabet. Which ones did he take?

Four letters from the alphabet are on the screen. You type in the missing letter. The computer will keep score. How many can you get right before time is up and the game ends? (*Hint:* Sing the alphabet song if you get stuck.)

For kids who know the letter names.

Missing Program

```
1  ? " ↰ TYPE IN THE MISSING LETTER.
2  FOR T=1 TO 1800:NEXT T
3  POKE 18,0:POKE 19,0:S=0:OPEN #1,4,0,"K:"
4  GR. 18:TRAP 4
5  M=INT(5*RND(0)):L=INT(21*RND(0)+65)
6  IF PEEK(19)>20 THEN 18
7  POS. 3,1:POKE 764,255
8  FOR C=0 TO 4
9  ? #6;CHR$(L+C);:NEXT C
10  POS. 3+M,1:? #6;"—"
11  IF PEEK(764)=255 THEN 11
12  POS. 3+M,1:? #6;CHR$(L+M+32):GET #1,A
13  IF A=L+M THEN FOR T=1 TO 60:SO. 1,29,10,8:NEXT T:S=S+1:POS. 9,6:? #6;S
14  POS. 3+M,2:? #6;CHR$(A)
15  IF A< >L+M THEN FOR T=1 TO 60:SO. 1,200,12,8:NEXT T
16  SO. 1,0,0,0
17  FOR T=1 TO 1200:NEXT T:POS. 3+M,2:? #6;"  ":GOTO 5
18  FOR F=1 TO 200:SO. 1,F*98,10,4:NEXT F:SO. 1,0,0,0
19  POS. 5,8:? #6;"GAME OVER
20  FOR T=1 TO 3000:NEXT T:GOTO 3
```

Variations

To make the game longer change line 6.

Old line 6: IF PEEK(19) > 20 THEN 18

New line 6: IF PEEK(19) > 40 THEN 18

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Turn TV volume up.

How It Looks



4. Type in the letter that is missing.
5. Press RETURN



6. This answer was right. A happy sound plays if you're right. A razzberry if you're not.



7. The next game starts immediately.



8. To end, press **SYSTEM RESET**

9. Save on disk or cassette as *Missing*.

Tally Sheet

How many stolen letters did you get back?

Name	*Day*	*Score*	*Name*	*Day*	*Score*

Remember The Next Letter

Watch the screen and you will see four letters appear on the screen, one at a time. Each one will disappear, so you'll have to remember them in order. Type in the first letter. Then the next letter. Then the next. (*Hint:* Say the letters out loud as they appear to help remember them.)

If you're right the computer shows RIGHT and you get one point. If you are wrong, it shows NOPE.

For kids who can read the alphabet.

Next Program

```
1  DIM L$(4),A$(4)
2  FOR C=1 TO 4
3  GR. 2
4  POS. 5+C,1
5  L$(C,C)=CHR$(21*RND(0)+65)
6  ? #6;L$(C,C)
7  FOR T=1 TO 600:NEXT T
8  NEXT C
9  GR. 2
10  ? "TYPE IN ALL FOUR LETTERS.
11  ? "THEN PRESS RETURN
12  POKE 764,255
13  INPUT A$
14  IF L$=A$ THEN D=10:GOTO 16
15  D=8
16  POSITION 3,2:? #6;L$
17  POSITION 3,3:? #6;A$
18  FOR T=1200 TO 1 STEP -2:SOUND  1,T,D,10
19  NEXT T:SOUND 1,0,0,0
20  GOTO 2
```

Variations

To make the letters stay on the screen longer, change line 7.

Old line 7: FOR T = 1 TO 600:NEXT T

New line 7: FOR T = 1 TO 1200:NEXT T

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up TV volume.
3. Type RUN.
4. Four letters will appear, one at a time.

How It Looks





5. Type in all four letters in the same order. Then press **RETURN** .



6. A happy siren tells you that you're right. A grouchy siren means "Nope!"



7. To end, press **SYSTEM RESET** .
8. Type RUN.
9. Save on disk or cassette as *Next*.

Reverse

Someone has been messin' with my favorite letters! They reversed them. Four letters appear on the screen, one at a time. Can you remember them in the *reverse* order? It's really hard!

Watch the screen and you will see four letters appear on the screen, one by one. Each one will disappear, so you'll have to remember them. Now type them in using the *reverse* order. Type in the fourth letter . . . then the third . . . then the second . . . then the first.

If you're right, the computer makes a happy sound; give yourself one point. If you are wrong, it makes an unhappy sound.

Reverse Program

```
1  DIM L$(4),A$(4)
2  FOR C=4 TO 1 STEP-1
3  GR. 2
4  POS. 5-C,1
5  L$(C,C)=CHR$(21*RND(0)+65)
6  ? #6;L$(C,C)
7  FOR T=1 TO 600:NEXT T
8  NEXT C
9  GR. 2
10  ? "TYPE IN ALL FOUR LETTERS
11  ? "IN REVERSE ORDER.
12  ? "THEN PRESS RETURN
13  POKE 764,255
14  INPUT A$
15  IF L$=A$ THEN D=10:GOTO 17
16  D=8
17  POS. 3,2:? #6;L$
18  POS. 3,3:? #6;A$
19  FOR T=1200 TO 1 STEP-2:SO.1,T,D,10
20  NEXT T:SO. 1,0,0,0:GOTO 2
```

Variations

To make the letters stay on the screen longer, change line 7.

Old line 7: FOR T = 1 TO 600:NEXT T

New line 7: FOR T = 1 TO 1200:NEXT T

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Turn up TV volume.
4. Four letters will appear, one at a time, then disappear.

How It Looks









5. Type in all four letters in the *REVERSE* order. Then press RETURN .



6. A happy siren tells you that you're right. A grouchy siren means "Nope!"



7. To end, press SYSTEM RESET
8. Save on disk or casette as *Next*.

WORD GAMES

Pig Latin Words

The computer changes words to *Pig Latin*. To talk *Pig Latin*, the first letter of your word moves to the end of the word. Then the letters "AY" are added to the word. "PIG" changes to "IGPAY." What happens to your name?

For kids who can read.

Pig Word Program

```
1  DIM W$(100),P$(100),L$(1)
2  ? " ↰ "
3  ? "TYPE IN A WORD
4  TRAP 3
5  INPUT W$
6  W=LEN(W$)
7  P$=W$
8  L$=P$(1,1)
9  IF L$="A" OR L$="E" OR L$="I" OR L$="O" OR L$="U" THEN
   P$(W+1)="YAY":GOTO 13
10  P$(W+1)=W$(1,1)
11  P$(W+2)="AY"
12  P$=P$(2)
13  ? "YOUR WORD IN PIG LATIN IS  ";
14  ? P$
15  GOTO 3
```

Variations

You can make your own *Barnyard Latin* if you change line 11. Type any two letters where AY are.

Old line 11: Pig$(W + 2) = "AY

New line 11: Pig$(W + 2) = "(any two letters)

How It Works

1. Type in the program or load from disk or cassette.
2. Type RUN.
3. Type your word.
4. Press **RETURN** .

How It Looks

```
TYPE IN A WORD
? DOG
YOUR WORD IN PIG
LATIN IS OGDAY
TYPE IN A WORD
? ■
```

5. To end, press **SYSTEM RESET** .
6. Save on disk or cassette as *Pigword*.



Pig Talk

Pig Talk is a two-player game. Player 2 should close her eyes. Player 1 types in a grammatical English sentence with four words. The computer translates it into simple Pig Latin. After the fourth word, Player 2 can open her eyes. Player 2 hits the RETURN key and the Pig Latin then appears. Player 2 must read the sentence without stumbling or bumbling before the sentence disappears, and translate it. Then Player 2 types in a sentence. If you make a mistake typing, hit SYSTEM RESET, type RUN, and start over.

Pigtalk Program

```
1  OPEN #2,4,0,"K:"
2  ? " ↰ ";"TYPE IN YOUR SECRET MESSAGE
3  DIM PIG$(999), L$(1)
4  FOR N=1 TO 999
5  GET #2,K
6  SET. 1,8,4
7  L$=CHR$(K)
8  PIG$(N)=L$
9  IF K< >155 AND L$< >"  " THEN NEXT N
10  PIG$(N)=PIG$(1,1)
11  PIG$(N+1)="AY  "
12  ? PIG$(2);
13  IF K=155 THEN SET. 1,8,10
14  IF K=155 THEN FOR T=1 TO 3000:NEXT T:? " ↰ ":END
15  GOTO 4
```

Variations

If the message disappears too quickly, change line 14.

Old line 14: IF K=155 THEN FOR T=1 TO 3000:NEXT T:? " ":END
New line 14: IF K=155 THEN FOR T=1 TO 4500:NEXT T:? " ":END

If it doesn't disappear soon enough, change line 14 this way.

IF K=155 THEN T=1 TO 2400:NEXT T:? " ":END

Keep playing with the numbers so that it's right for you.

Secret Latin

You can make up your secret language. Just change line 11. Type any two letters where "AY " are.

Old line 11: PIG$(N + 1) = "AY
New line 11: PIG$(N + 1) = "IZ

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Player 2 should close her eyes.
4. Player 1 should type in a four-word sentence.

   If you make a mistake, hit **SYSTEM RESET** .

5. Player 2 should open her eyes.
6. Player 2 should press the **RETURN** key.
7. Player 2 should read message correctly before it disappears. Translate it.
8. Now it's Player 2's turn to type a message. Type RUN and continue. Which of you is the piggy-est?
9. To end, press **SYSTEM RESET** .
10. To begin again, type RUN.
11. Save on disk or cassette as *Pigtalk*.

How It Looks

```
TYPE IN YOUR SECRET
MESSAGE.
```

```
TYPE IN YOUR SECRET MESSAGE
OURYAY ARROTPAY
ALKSTAY ELLWAY
```



Code

In this game you can send messages to your friends in a number code. How fast can your friend decode the message? How about you?

Player 1 types in a four-letter word. The computer will change the letters into numbers. Player 2 then translates the message back into words. (The numbers and letters at the bottom of the screen may help.) If you are right, you get a point. When you get to 10 points, the number clues at the bottom disappear and you're on your own. The first player to reach 20 wins.

Code Program

```
1  OPEN #2,4,0,"K:"
2  DIM A$(100),L$(100)
3  DIM S(2):S(1)=0:S(2)=0:P=1
4  ? " ↰ TYPE IN YOUR WORD, PLAYER  ";P
5  A$="  ":FOR C=1 TO 100
6  GET #2,X:IF X=155 THEN 8
7  A$(C,C)=CHR$(X):? X-64;"  ";:NEXT C
8  ? :? :? "DECODE IT":IF S(P)>9 THEN 14
9  POS. 2,20:L=65
10  FOR N=1 TO 26
11  ? CHR$(L);L-64;"/";
12  L=L+1
13  NEXT N:POS. 3,4
14  INPUT L$
15  IF L$< >A$ THEN ? "NOPE!":GOTO 17
16  ? "RIGHT!":S(P)=S(P)+1
17  ? "PLAYER 1      PLAYER 2
18  ? S(2),,S(1)
19  P=P+1:IF P> 2 THEN P=1
20  FOR T=1 TO 1800:NEXT T:POKE 764,255:GOTO 4
```

Variations

To make the game easier, you can change line 8 to leave the code on the screen.

Old line 8: ? :? :? : "DECODE IT":"IF S(P) > 9 THEN 14

New line 8: ? :? :? :"DECODE IT"

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Player 1 types in a four-letter word. It appears as numbers. In this example, *yuck* was used.
4. Press **RETURN** .
5. Player 2 decodes the numbers and types in the word.
6. Press **RETURN** .
7. You get 1 point if you are correct. After you score 10, the number clues disappear from the bottom of the screen. The first player to reach 20 wins.

How It Looks

```
TYPE IN YOUR WORD, PLAYER 1
```

```
TYPE IN YOUR WORD, PLAYER 1
25  21  3  11 ■
```

```
TYPE IN YOUR WORD, PLAYER 1
25  21  3  11
DECODE IT
? ■

A1/B2/C3/D4/E5/F6/G7/H8/I9/J10/
K11/L12/M13/N14/O15/P16/Q17/R18/
S19/T20/U21/V22/W23/X24/Y25/Z26
```

```
TYPE IN YOUR WORD, PLAYER 1
25  21  3  11
DECODE IT
? YUCK
RIGHT!
PLAYER 1          PLAYER 2
   0                 1

A1/B2/C3/D4/E5/F6/G7/H8/I9/J10/
K11/L12/M13/N14/O15/P16/Q17/R18/
S19/T20/U21/V22/W23/X24/Y25/Z26
```

8. To play again, press **SYSTEM RESET** .
9. Type RUN.
10. Save on disk or cassette as *Code*.

Tally Sheet

Name	*Day*	*Score*	*Name*	*Day*	*Score*



Grandmother's Trunk

There are tons and tons of things you can hide in Grandmother's trunk. Can you and your friends put them in and remember what you hid—in the right order? How many can you remember?

Player 1 types in the name of something that she is putting in Grandmother's trunk like "marbles." Player 2 types in "marbles" and "horses." The next player types in "marbles," "horses" and "pickles." As long as you can remember all the things in Grandmother's trunk, you can add to the list. As soon as you make a mistake, you're out of the game until it begins over again.

Trunk Program

```
1  DIM L$(100),P$(100),W$(100),R$(100)
2  C=0
3  C=C+1
4  POKE 764,255
5  ? " ↰ PUT SOMETHING IN GRANDMOTHER'S TRUNK"
6  INPUT W$
7  L$(LEN(L$)+1)=W$
8  P$=""
9  ? " ↰ REPEAT THE LIST SO FAR"
10  FOR A=1 TO C
11  INPUT R$
12  P$(LEN(P$)+1)=R$
13  FOR B=1 TO LEN(P$)
14  IF L$(B,B)< >P$(B,B) THEN ? "OOPS! YOU FORGOT SOMETHING!":END
15  NEXT B
16  NEXT A
17  ? "RIGHT!"
18  FOR T=1 TO 900:NEXT T
19  GR. 0
20  GOTO 3
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Type in the name of something you'd like to store.
4. Press RETURN .

How It Looks

```
PUT SOMETHING IN
GRANDMOTHER'S TRUNK

? CANDY ■
```

5. The next player types in "candy," for instance. Press RETURN after each item.

```
REPEAT THE LIST SO FAR

CANDY ■
```

6. If you're wrong, you get this friendly message.

```
OOPS! YOU FORGOT
SOMETHING!

READY
■
```

7. If you're right, the computer says so.

```
RIGHT!
```

8. Then *you* get to store something, too, like "fingers."

```
STORE SOMETHING IN
GRANDMOTHER'S TRUNK
? FINGERS ■
```

9. To start over type RUN.

10. Save on disk or cassette as *Trunk*.



Name Games

Family Tree

Watch the computer make a family tree from your family's names.

Tree Program

```
1  DIM N$(38),F$(38),M$(38)
2  ? " ↰ WHAT IS YOUR NAME?
3  INPUT N$
4  ? "MOTHER'S?
5  INPUT M$
6  ? "FATHER'S?
7  INPUT F$
8  L=LEN(M$)
9  L1=L+LEN(N$)+2
10  M$(L+1)="+"
11  M$(L+2)=N$
12  M$(L1)="+"
13  M$(L1+1)=F$
14  GR. 0
15  SET. 2,12,4
16  FOR C=1 TO LEN(M$)
17  POS. 20-C,C
18  ? M$(1,C);M$(1,C)
19  NEXT C
```

How It Works

1. Type in program or load it from disk or cassette.
2. Type RUN.

How It Looks



3. Type in your name. Press RETURN after each name.



4. Type in your mother's name.



5. Type in your father's name.

```
WHAT IS YOUR NAME?
STEVE
MOTHER'S?
?MOM
FATHER'S?
?DAD
```

6. The computer will make a family tree from your family's names.

```
                                MM
                               MOMO
                              MOMMOM
                             MOM+MOM+
                            MOM+SMOM+S
                           MOM+STMOM+ST
                          MOM+STEMOM+STE
                         MOM+STEVMOM+STEV
                        MOM+STEVEMOM+STEVE
                       MOM+STEVE+MOM+STEVE+
                      MOM+STEVE+DMOM+STEVE+D
                     MOM+STEVE+DAMOM+STEVE+DA
                    MOM+STEVE+DADMOM+STEVE+DAD
```

7. To do again, type RUN.

8. Save on disk or cassette as *Tree*.

Flash

Have you ever seen your name in flashing lights? Type in your first name. You will see your name flash on the screen.

Flash Program

```
1  DIM A$(30)
2  ? " ↰ TYPE IN YOUR NAME
3  INPUT A$
4  GR. 18
5  POS. 6,5
6  ? #6;A$
7  FOR N=1 TO 15
8  SET. 0,N,N
9  FOR T=1 TO 20
10  NEXT T
11  NEXT N
12  GOTO 7
```

Variations

If you change line 2, you can have any word or name flash on the screen.

Old line 2: ?"TYPE IN YOUR NAME

New line 2: ?"TYPE IN YOUR FRIEND'S NAME

or

New line 2: ?"WHAT IS YOUR FAVORITE ANIMAL?

How It Works

1. Type in program or load it from disk or cassette.
2. Type RUN.
3. Type in your name.
4. Press **RETURN** .
5. Press **SYSTEM RESET** to end.
6. To begin again, type RUN.
7. Save on disk or cassette as *Flash*.

How It Looks







Graffiti

Fool's names and fool's faces
Are often seen in public places . . .
like a television screen.

You can use the joystick to write your name. You can use your brother's, sister's, or friend's name, too. (*Note:* This program is based on the one called *Joystick Drawing*.)

Graffiti Program

```
1  DIM A$(30)
2  ? " ↰   TYPE NAME
3  ? "MOVE STICK TO DRAW
4  INPUT A$
5  GR. 0
6  POS. 20,10
7  ? "  ";
8  J=STICK(0)
9  IF J=7 THEN ? A$;
10  IF J=11 THEN ? "←";A$;"←";
11  IF J=14 THEN ? "↑";A$;"↑";
12  IF J=13 THEN ? "↓";A$;"↓";
13  GOTO 8
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Type in your name.

How It Looks



4. Press **RETURN** .



5. Move joystick to "write."



6. Press **SYSTEM RESET** to end.
7. To begin again, type RUN.
8. Save on disk or cassette as *Graffiti*.

Hello

The computer gives a tremendous hello to you and your friends.

Hello Program

```
1  DIM A$(15)
2  ? " ↰   WHAT IS YOUR NAME?
3  INPUT A$
4  GR. 23:COL. 1
5  A=155:B=5:C=2
6  PLOT A,B
7  A=A+2*C
8  SO. 0,A,14,8
9  DRAWTO A,B
10  C=C-(2*C)
11  B=B+C
12  DRAWTO A,B
13  IF C>0 THEN C=C+2
14  C=C-3
15  IF B<70 THEN 7
16  FOR D=1 TO 100:SO. 3,D,6,10
17  GR. 18:POS. 0,8
18  ? #6;"HELLO,  ";A$:NEXT D
19  FOR T=1 TO 20:NEXT T
20  GOTO 4
```

Variations

You can change the message by changing the words in lines 2 and 18.

Old line 2: ? " ↰ WHAT IS YOUR NAME?

New line 2: ? " ↰ WHAT IS YOUR PET'S NAME?

Old line 18: ? #6;"HELLO, ";A$:NEXT D

New line 18: ? #6;"HERE'S A LION, ";A$:NEXT D

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up the TV volume a little.
3. Type RUN.
4. Type in your name and press **RETURN**.

How It Looks



5. The screen looks like this . . .



6. Press **SYSTEM RESET** button to start again.
7. Type RUN.
8. Save on disk or cassette as *Hello*.

People Chase

Is your name dull? Brighten it up with this computer program. The computer will chase your name across the screen.

Chase Program

```
1  DIM N$(30)
2  X=0
3  Y=0
4  ? " ↰ TYPE IN YOUR NAME
5  INPUT N$
6  GR. 18
7  X=X+1
8  IF X>19 THEN X=0:Y=Y+1
9  IF Y>9 THEN END
10  POS. X,Y
11  ? #6;N$
12  GOTO 6
```

Variations

You can type in your friend's name or messages like "Happy Birthday" or "Go Home." You can also type in symbols:

Press the ESC key.

Hold down the CTRL key and type a letter key at the same time.

How It Works

1. Type in program or load it from disk or cassette.
2. Type RUN.

How It Looks



3. Type in your name.
4. Press RETURN .



5. Your name will be chased across the screen.





6. To begin again, type RUN.

7. Save on disk or cassette as *Chase*.

Rollcall

With this program the computer prints your name one letter at a time. Your name rolls down the screen. Press **BREAK** to stop.

Rollcall Program

```
1  DIM A$(40)
2  ? " ↰ WHAT IS YOUR NAME
3  INPUT A$
4  TRAP 4
5  SET. 2,5,4
6  FOR C=1 TO 40
7  ? A$(1,C)
8  NEXT C
```

Variations

You can type in your first, middle and last names! Try a long name. Now try a short one. If you want the computer to ask you to name some animals, change line 2. For example,

Old line 2: ? " ↰ WHAT IS YOUR NAME?

New line 2: ? " ↰ WHAT IS THE LONGEST WORD YOU KNOW?

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.

How It Looks

```
WHAT IS YOUR NAME?
? ■
```

3. Type in your name.

```
WHAT IS YOUR NAME?
? RUDOLPH ■
```

4. Press **RETURN** .

```
WHAT IS YOUR NAME?
? RUDOLPH
R
RU
RUD
RUDO
RUDOL
RUDOLP
RUDOLPH
WHAT IS YOUR NAME
? RUDOLPH THE RED NOSED
REINDEER
```

5. Press the **BREAK** key to stop.
6. To run again, type RUN.
7. To end, press **SYSTEM RESET** .
8. Save on disk or cassette as *Rollcall*.

Number Games

Teepees

Look deep into the dark.
Soon there will be a spark.
How many TEEPEES do you see?
Count them very carefully.

Teepees will flash on the screen. Then they will disappear. Count them and type in the number. If you are right, the computer will show YES. If you are wrong, the computer will show NO and tell you the right number. *(Hint:* If it's hard, point to each teepee and count out loud.) *Note:* Sometimes the teepees will appear on top of each other. Listen to the sounds.

For kids who can add to 6.

Teepees Program

```
1  GR. 2
2  TRAP 1
3  N=INT(2+RND(0)*5)
4  FOR C=1 TO N
5  X=RND(0)*18
6  Y=RND(0)*9
7  POS. X,Y
8  ? #6;"^"
9  SO. 1,80,10,4
10  FOR T=1 TO 200:NEXT T
11  SO. 1,0,0,0
12  NEXT C
13  FOR T=1 TO 1800:NEXT T
14  POKE 764,255
15  GR. 2:? "HOW MANY TEEPEES?
16  INPUT A
17  IF A=N THEN ? #6;"RIGHT!":GOTO 19
18  ? #6;"NO . . . ";N
19  FOR T=1 TO 1200:NEXT T
20  GOTO 1
```



How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up TV volume.
3. Type RUN.

How It Looks





4. Type in the number of teepees you saw.
5. Press RETURN .



6. The computer will tell you if you're right or not . . .



7. More teepees will flash on the screen.
8. To end, press SYSTEM RESET .
9. Save on disk or cassette as *Teepee*.

Variations

EASIER: To make the game easier, there are several things that can be done.

1. To make the teepees stay on the screen longer so that the child can count them, increase the amount of time in line 13.

   Old line 13: FOR T=1 TO 1800:NEXT T
   New line 13: FOR T=1 TO 2400:NEXT T

2. If the child cannot add to 6, lower the number of teepees possible. For example, change line 3 so that up to three teepees will appear.

   Old line 3: N=INT(2+RND(0)*5)
   New line 3: N=INT(2+RND(0)*2)

HARDER: To make the game harder, there are several things that can be done.

1. To make the teepees disappear sooner so that the child can only glance at them, decrease the amount of time in line 13.

   Old line 13: FOR T=1 TO 1800:NEXT T
   New line 13: FOR T=1 TO 600:NEXT T

2. Raise the number of teepees in line 3.

   Old line 3: N=INT(2+RND(0)*5)
   New line 3: N=INT(2+RND(0)*10)

Note: Fine tune these numbers till the activity fits the child's ability and attention span.

Scoop Up The Numbers

Scoop up the numbers 1 through 9 in the right order. The numbers 1 through 9 explode on the screen. Move the joystick around as quickly as you can from 1 to 9. Use the red button to scoop up the number that you need. A bell means "You got it!" A razzberry noise means "Try Again!" *Hint:* Tap stick for more control.

For kids who are learning to count to 10.

Scoop Program

```
1  OPEN #2,12,0,"S:"
2  ? " ↰ PICK UP NUMBERS 1 TO 9
3  ? "PUSH BUTTON TO PICK UP NUMBER
4  FOR T=1 TO 900:NEXT T
5  ? " ↰ ":FOR C=1 TO 9
6  H=32*RND(0)+2:V=2*C
7  FOR A=1 TO 3:FOR B=1 TO 3
8  POS. H+A,V+B:? C;
9  NEXT B:NEXT A:NEXT C:N=177
10  J=STICK(0)
11  IF N=186 THEN ? :? "YOU'RE THE CHAMP!":END
12  IF J=11 THEN ? "←";
13  IF J=7 THEN ? "→";
14  IF J=14 THEN ? "↑";
15  IF J=13 THEN ? "↓";
16  IF STRIG(0)=1 THEN 10
17  GET #2,A:IF A< >N THEN D=6:GOTO 19
18  N=N+1:? " ◁ ";:D=10
```

Note: To type this character, press ESC once and then press DELETE BACK S.

```
19  FOR T=1 TO 100:SO. 0,150,D,8
20  NEXT T:SO. 0,0,0,0:GOTO 10
```

Variations

To make the game easier you can have the numbers go up to 3 rather than 9. Change lines 5 and 11.

Old line 5: ? " ↰ ":FOR C=1 TO 9

New line 5: ? " ↰ ":FOR C=1 TO 3

Old line 11: IF N=186 THEN ? :?"YOU'RE THE CHAMP!":END

New line 11: IF N=180 THEN ? :?"YOU'RE THE CHAMP!":END

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up the TV volume.
3. Type RUN.
4. Plug joystick into jack 1.
5. The screen will look something like this:

How It Looks



6. Move white cursor over a 1.
7. Press the red trigger button on joystick.
8. Move to a 2 and so on till you scoop 9.
9. When you get to 9, you become the scoop champ!



10. To play again, type RUN.
11. Save on disk or cassette as *Scoop*.

Diamond

An old miner is looking for diamonds. Sometimes he finds them. Sometimes he finds glass. No one will buy glass. Your job is to help him find diamonds.

The computer will build a shape with many sides. Your job is to count how many sides there are by the time the buzzer stops ringing. You type in your answer. If you are right, the number of sides is added to your loot score. If you are wrong, your answer is subtracted from your loot score. Hope you end up with a lot of diamonds and no glass!

For kids who can add to 7. (If you want a version where a little kid can play against a bigger kid, see the *Bigger* game.)

Diamond Program

```
1  GR. 7
2  DEG
3  S=0
4  FOR C=1 TO 10
5  N=INT(RND(0)*4+3)
6  FOR D=0 TO 360 STEP 360/N
7  COLOR 1
8  PLOT SIN(D)*45+78,COS(D)*34+38
9  DRAWTO SIN(360/N+D)*45+78,COS(360/N+D)*34+38
10  NEXT D
11  FOR T=1 TO 350
12  SO. 1,T*98,10,4:NEXT T
13  GR. 7:POKE 764,255
14  SO. 1,0,0,0:TRAP 8
15  ? "HOW MANY SIDES";:INPUT A
16  IF A=N THEN ? "DIAMONDS!":S=S+A:GOTO 18
17  ? "GLASS! THERE ARE  ";N;" SIDES":S=S-ABS(N-A)
18  ? "YOUR LOOT IS WORTH ";S
19  FOR T=1 TO 600:NEXT T:NEXT C
20  ? "GAME OVER
```

Variations

If you want to make the game harder, change line 11. The picture disappears sooner.

Old line 11: FOR T=1 TO 350

New line 11: FOR T=1 TO 200

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up the TV volume.
3. Type RUN.
4. A shape appears.

How It Looks



5. Type in the number of sides.
6. Press RETURN .



7. This player typed in 4.



8. He was right.



9. The next stone will appear. How many diamonds can you own?

10. To end the game, press **SYSTEM RESET**.

11. Save on disk or cassette as *Diamond.*

Tally Sheet

How many diamonds?

Name	*Day*	*Score*	*Name*	*Day*	*Score*

Bigger

Bigger is a variation on *Diamond*. In this game, however, a big kid (like a parent, brother or sister) can play against a little kid. The game has a built-in handicap. The big kid gets harder problems to solve and receives a handicapped score.

A shape will appear on the screen. You have to count how many sides it has before it disappears. Type in the answer. The first player who reaches a score of 25 wins!

Shape 1 appears. Player 1 counts how many sides it has. He types in the number. If it is correct, the number of sides is added to his score. Shape 2 appears. Player 2 takes her turn. It will have up to fifteen sides. If the answer is not correct, the difference between the number of sides the shape has and the number the player typed is subtracted from the score. The scores are handicapped so that the big kid's score is proportional to the little kid's score.

For little kids who can add and subtract up to 6 and big kids who can add to 15. The little kid should be player 1 and the older one should be player 2.



Bigger Program

```
1  DEG :DIM S(2):S(1)=0:S(2)=0:P=1
2  GR. 7
3  IF P=1 THEN N=INT(RND(0)*4+3)
4  IF P=2 THEN N=INT(RND(0)*5+11):IF N<12 THEN 3
5  TRAP 15:FOR D=0 TO 360 STEP 360/N
6  COL. C:C=C+1:IF C=4 THEN C=1
7  PL. SIN(D)*45+78,COS(D)*34+38
8  DRAWTO SIN(360/N+D)*45+78,COS(360/N+D)*34+38
9  NEXT D
10  FOR T=1 TO 300:NEXT T
11  FOR T=1 TO 350:SO. 1, T*98,10,4:NEXT T:SO. 1,0,0,0
12  GR. 7:POKE 752,1:? "PLAYER  ";P;":HOW MANY SIDES?
13  POKE 764,255:TRAP 110:INPUT A
14  IF A=N THEN ? "RIGHT!";:S(P)=S(P)+A:IF P=2 THEN S(P)=S(P)-9:GOTO 16
15  IF A< >N THEN ? "NO!";:S(P)=S(P)-ABS(N-A)
16  ? "  THERE ARE ";N;"  SIDES
17  ? "PLAYER 1        PLAYER 2":POKE 752,1
18  ? S(1),,S(2)
19  P=P+1:IF P>2 THEN P=1
20  FOR T=1 TO 1800:NEXT T:GOTO 2
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up TV volume.
3. Type RUN. A shape appears.

How It Looks



4. Player 1 (the little kid) should type in the number of sides.



5. This player typed in 4.



6. Press **RETURN** .



7. Another shape appears.



8. Player 2 (the big kid) should type in the number of sides.

```
PLAYER 2: HOW MANY
SIDES? ■
```

9. Player 2 will get a "handicapped" score.

```
PLAYER 2: HOW MANY
SIDES? 13
```

10. The first player to get 25 points wins the game!

11. Press **SYSTEM RESET** to end game.

12. To begin again, type RUN.

13. Save on disk or cassette as *Sides.*

Variations

To make the game easier, raise the length of time the shape stays on the screen by changing line 10.

Old line 10: For T=1 TO 300:NEXT T

New line 10: For T=1 TO 900:NEXT T

To make the game harder, lower the length of time the shape appears by changing line 10.

Old line 10: For T=1 TO 300:NEXT T

New line 10: For T=1 TO 150:NEXT T

Tally Sheet

Who got 25 points first?

Name	*Day*	*Score*	*Name*	*Day*	*Score*

Count the Sides

Count the Sides is a two-player game. It is harder than *Diamond.* Shape 1 appears. Player 1 counts how many sides it has. He types in the number. If he is correct, the number of sides is added to his score. Shape 2 appears. Player 2 takes her turn. If the answer is not correct, the difference between the number of sides the shape has and the number the player typed is subtracted from the score. The first player to reach a score of 50 wins!

For kids who can add and subtract up to 15.

Sides Program

```
1  DEG :DIM S(2):S(1)=0:S(2)=0:P=1
2  GR. 7
3  N=INT(RND(0)*12+3)
4  TRAP 2:FOR D=0 TO 360 STEP 360/N
5  COL. C:C=C+1:IF C=4 THEN C=1
6  PLOT SIN(D)*45+78,COS(D)*34+38
7  DRAWTO SIN(360/N+D)*45+78,COS(360/N+D)*34+38
8  NEXT D
9  FOR T=1 TO 600:NEXT T
10  FOR T=1 TO 60
11  SO. 1,200,12,8:NEXT T
12  SO. 1,0,0,0
13  GR. 7:? "PLAYER ";P;":HOW MANY SIDES";:INPUT A
14  IF A=N THEN ? "RIGHT!";:S(P)=S(P)+A:GOTO
15  ? "NO!";:S(P)=S(P)-ABS(N-A)
16  ? " THERE ARE ";N;" SIDES
17  ? :? "PLAYER 1       PLAYER 2
18  ? S(1),,S(2),
19  P=P+1:IF P>2 THEN P=1
20  FOR T=1 TO 1800:NEXT T:GOTO 2
```



Variations

To make the game easier, lower the number of sides by changing line 3.

Old line 3: N=INT(RND(0)*12+3)

New line 3: N=INT(RND(0)*5+3)

To make game harder, lower the length of time the shape appears by changing line 9.

Old line 9: FOR T=1 TO 600:NEXT T

New line 9: FOR T=1 TO 300:NEXT T

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up the TV volume.
3. Type RUN.

How It Looks



4. Player 1 should type in the number of sides.



5. This player typed in 4.



6. Press **RETURN** .

```
RIGHT! THERE ARE 4
SIDES.
PLAYER 1        PLAYER 2
    4               0
```

7. Another shape appears.



8. Player 2 should type in the number of sides.

```
PLAYER 2: HOW MANY
SIDES? ■
```

9. This player typed in 7.

```
PLAYER 2: HOW MANY
SIDES? 7 ■
```

10. The computer subtracts 7 from 0 and player 2 is left with -1.



11. The first player to get 50 points wins the game!

12. Press **SYSTEM RESET** to end game.

13. To begin again, type RUN.

14. Save on disk or cassette as *Sides.*

Tally Sheet

Who got 50 points first?

Name	*Day*	*Score*	*Name*	*Day*	*Score*

How Many X's??

The screen fills up with X's. How many do you think are there? How many guesses does it take you to get the right number? How long does it take your friends?

Watch the screen as it fills up. Each time it fills up, there will be a different number. Type in the number of X's that you think you've seen. The computer will tell you if there are more or less than you guess. Just guess!

For kids who can "count" to 200.

X Program

```
1  GR. 18
2  R=0
3  N=INT(50+RND(0)*200)
4  FOR C=1 TO N
5  X=RND(0)*18
6  Y=RND(0)*11
7  POS. X,Y
8  ? #6;"X"
9  NEXT C
10  GR. 0
11  ? "HOW MANY X'S?
12  TRAP 6
13  R=R+1
14  ? "ROUND ";R
15  INPUT A
16  IF A=N THEN ? "THAT'S RIGHT!":TRAP 1:INPUT A
17  IF A<N THEN ?"NO.. THERE ARE MORE ":GOTO 11
18  IF A>N THEN ? "NO.. FEWER THAN THAT":GOTO 11
```

Variations

To make this easier, decrease the numbers in line 3. This will make between 20 and 50 X's appear.

Old line 3: N=INT(50+RND(0)*200)

New line 3: N=INT(20+RND(0)*30)

How It Works

1. Type in program or load it from disk or cassette.
2. Type RUN.

How It Looks



3. Type in the number of X's you think are on the screen. Just guess!

```
HOW MANY X'S?
ROUND 1
? ■
```

4. This player typed 7.
5. Press **RETURN** .

```
HOW MANY X'S?
? 7
NO. THERE ARE MORE.
```

6. Type in another number—either bigger or smaller—depending on what the computer says.

```
HOW MANY X'S?
ROUND 2
200
NO...THERE ARE FEWER.
```

7. Continue typing in your guesses until you're right. How many rounds did it take you?

```
? 48
THAT'S RIGHT
```

8. The screen will fill up with X's again. Then you or your friend can start guessing how many X's are on the screen now.



9. Press **SYSTEM RESET** to end.

10. Save on disk or cassette as *X*.

Tally Sheet

What's your lowest score?

Name	Day	Score	Name	Day	Score

Pie

How many pieces of pie do you want sliced? Are you very, very hungry or do you have a lot of friends?

You type in the number of pieces. The computer does the rest. Is it always right? Count the pieces and see!

For kids who can count.

```
1  GR. 7
2  COL. 1
3  POKE 752,1
4  DEG
5  TRAP 5
6  ? "HOW MANY SLICES SHOULD I CUT";
7  INPUT A
8  FOR B=0 TO 360 STEP 8
9  PL. 32*SIN(B)+80,32*COS(B)+40
10  NEXT B
11  FOR C=0 TO 360 STEP 360/A
12  FOR D=0 TO 16 STEP 2
13  E=D*SIN(C):F=D*COS(C)
14  PLOT 2*E+80,2*F+40
15  SO. 0,10*D,10,8:NEXT D
16  NEXT C
```

How It Works

1. Type in program or load it from disk or cassette.
2. Turn up the TV volume.
3. Type RUN.
4. You type in the number.
5. Press RETURN .

How It Looks



6. This kid asked for 10!



7. To do again, type RUN.
8. Save on disk or cassette as *Pie*.

Stoppit!

A number between 1 and 9 marches across the screen. You must STOPPIT! Type in the number before it goes off the screen. A score appears that tells how long it took you. The lower your score, the better. The game ends in ten rounds.

Stoppit Program

```
1  S=0
2  OPEN #2,4,0,"K:"
3  FOR C=1 TO 11
4  N=INT(9*RND(0))
5  GR. 0
6  POKE 752,1
7  POS. 3,3:? N
8  FOR H=0 TO 34
9  POS. 3,3
10  ? " ▶":X=0
11  POS. 10,10:? "STOPPIT TIME...";H
12  FOR T=1 TO 100:NEXT T
13  IF PEEK(764)=255 THEN 16
14  POS. 3,5:GET #2,X
15  ? CHR$(X)
16  IF N< >X-48 THEN NEXT H:H=H-1
17  S=S+H
18  POS. 10,18:? "SCORE...";S
19  FOR T=1 TO 900:NEXT T
20  NEXT C:? "GAME IS OVER
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. A number between 1 and 9 marches across the screen.

How It Looks



4. STOPPIT! You type in the number.



5. If you are right, the number stops. If you are wrong, it keeps going. You can try again.
6. Your score appears. It tells you how long it took you to stop the number. The lower the score, the better.



7. The game ends in ten rounds.
8. To play again, type RUN.
9. Save on disk or cassette as *Stoppit*.

Tally Sheet

How fast are you? The lower the score, the better.

Name	*Day*	*Score*	*Name*	*Day*	*Score*

Stop 'Em

This is a harder version of *Stoppit*. A number between 10 and 99 marches across the screen. You must type in the number before it goes off the screen. A score appears that tells you how long it took you. The lower your score, the better. The game ends in ten rounds.

Stoppem Program

```
1  S=0:OPEN #2,4,0,"K:"
2  FOR C=1 TO 11:N1=INT(9*RND(0)):N2=INT(9*RND(0)):GR. 0
3  POKE 752,1:POKE 764,255
4  POS. 3,3:? N1;N2
5  FOR H=1 TO 30
6  POS. 5,20:? "STOP 'EM TIME..";H
7  POS. 1,3:? " ▶"
8  FOR T=1 TO 50:NEXT T
9  IF PEEK(764)=255 THEN NEXT H:Z=30:GOTO 17
10  GET #2,X1:? CHR$(X1);:IF N1< >X1-48 THEN NEXT H
11  FOR Z=H TO 30
12  POS. 20,20:? Z
13  POS. 1,3:? " ▶":FOR T=1 TO 50:NEXT T
14  IF PEEK(764)=255 THEN 16
15  GET #2,X2:POS. 3,4:? CHR$(X2)
16  IF N2< >X2-48 THEN NEXT Z:Z=Z-1
17  S=S+Z
18  POS. 24,20:? "SCORE..";S
19  FOR T=1 TO 1200:NEXT T
20  NEXT C:? "GAME IS OVER
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. A number between 10 and 99 marches across the screen.

How It Looks



4. You type in the number.
5. If you are right, the number stops. If either number is wrong, the whole number keeps going. You can try again.
6. Your score appears. It tells you how long it took you to stop the number. The lower the score, the better.



7. The game ends in ten rounds.
8. To play again, type RUN.
9. Save on disk or cassette as *Stoppem*.

Tally Sheet

How long did it take you to Stop 'Em? Remember, the lower score wins!

Name	*Day*	*Score*	*Name*	*Day*	*Score*

Designs

Hearts

Hearts move and sing.

Move the joystick from side to side. Push the red button sometimes. If you push the red button and move the joystick at the same time, you can hear different sounds.

Hearts Program



```
1  ? " ↰ MOVE THE JOYSTICK
2  ? "PRESS RED BUTTON SOMETIMES
3  FOR T=1 TO 1200
4  NEXT T
5  GR. 0
6  SET. 2,3,2
7  POKE 752,1
8  IF STICK(0)=11 AND H>5 THEN H=H-1
9  IF STICK(0)=7 AND H<38 THEN H=H+1
10  POS. 0,23
11  ? "↓"
12  POS. H,23
13  ? "♥"
```

Note: To get this character, hold down CTRL and press ,

```
14  SO. 1,H,10,8
15  IF STRIG(0)=0 THEN SO. 1,H,6,8
16  GOTO 8
```

Variations

If you would like to change the heart to some other sign, change the letter in line 13.

Old line 13: ? " ♥ "

New line 13: ? " ● "

Any key can be used instead of , or T .

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up TV volume.
3. Type RUN.

How It Looks



4. Move the joystick from side to side. Press the red button. Move the joystick and press the red button at the same time.



5. Press **SYSTEM RESET** to end.
6. To begin again, type RUN.
7. Save on disk or cassettes as *Hearts*.

Snowflake

You can design your own snowflakes. Just type in a number between 3 and 10. What number gives the most fancy one?

Flake Program

```
1  GR. 8
2  COL. 1
3  ? " ↰   HOW FANCY(3-10)
4  INPUT A
5  FOR C=1 TO 360 STEP A
6  PLOT SIN(C)*45+78,COS(C)*30+38
7  DRAWTO SIN(C+A)*45+78,COS(C+A)*30+38
8  NEXT C
```



How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Type in a number like 4. Press **RETURN**

How It Looks



4. It should look like this . . .



5. Type RUN. Try another number.
6. To end, press **SYSTEM RESET** .
7. Save on disk or cassette as *Flake*.

Tunnel

The prisoners in jail have been digging lots of tunnels. You can see all the different kinds with this program.

You type in a number between 40 and 120 and a tunnel appears. Try 90. Try 89. See the difference?

Tunnel Program

```
1  GR. 8
2  DEG
3  COL. 1
4  TRAP 1
5  SET. 2,0,0
6  ? "WHAT KIND OF TUNNEL(40-120)?
7  INPUT N
8  FOR A=1 TO 75
9  PLOT SIN(N*A)*A+155,COS(N*A)*A+78
10  DRAWTO SIN(N*(A+1))*A+155,COS(N*(A+1))*A+78
11  NEXT A
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Type in the number of the tunnel you want to see. Press **RETURN** .

How It Looks



4. Your screen will look like this:



5. Type RUN.
6. Try another number like 89.



7. To end, hit **SYSTEM RESET** .
8. Save on disk or cassette as *Tunnel.*

Stars

This program makes brightly colored star designs. You decide how bright they should be.

Stars Program

```
1  ? " ↰   HOW BRIGHT(3-10)
2  INPUT A
3  GR. 23
4  SET. 4,6,4
5  FOR C=1 TO 360 STEP A
6  COL. C
7  FOR T=1 TO 10
8  SO. 1,60,8,10
9  NEXT T
10  SO. 1,0,0,0
11  PLOT SIN(C)*45+78,COS(C)*30+38
12 DRAWTO SIN(C+A)*45+78,COS(C+A)*30+38
13 NEXT C
```



How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up your TV volume a little.
3. Type RUN.
4. Type in a number between 3 and 10 for how bright you want the star.

How It Looks



5. Press RETURN .
6. The screen should look like this:



7. Type RUN.
8. Try another number between 3 and 10.
9. Save on disk or cassette as *Stars*.

Colors

Colors

This listing can help you program. It cycles through the colors and prints the SETCOLOR statement that makes it.

When you run the program, the background starts out with color 0 and the darkest shade (which is black). It prints the SETCOLOR statement being used. You can use the SETCOLOR statement you like in your next program. Each time you hit the RETURN key, the shade lightens through all color combinations.

Colors Program

```
1  DIM A$(1)
2  FOR H=0 TO 15
3  FOR L=1 TO 14
4  GR. 0
5  SET. 2,H,L
6  ? "SETCOLOR 2,";H;",";L
7  ? "HIT RETURN KEY FOR NEXT COLOR
8  INPUT A$
9  NEXT L
10  NEXT H
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. The first line is the SETCOLOR statement in GRAPHICS 0.
4. Hit RETURN to see next shade. Hold RETURN down to flip through.

How It Looks



Piemaker

This is a silly baking program. How well do you know your flavors? For example, you are asked what color a spinach pie is! If you're right, the computer makes the pie and plays a pretty melody. If you're wrong, the computer makes it anyway, but plays a yucky melody!

Piemaker Program

```
1  DIM C$(10),P$(10),A$(30),I$(1)
2  DEG :RESTORE :FOR N=1 TO 6
3  DATA G, SPINACH,11,5,P,PLUM,7,2
4  DATA Y,LEMON,1,12,0,CARROT,2,8
5  DATA R,PIZZA,3,2,B,BLUEBERRY,8,7
6  READ C$,P$,H,L
7  ? " ↰  LET'S MAKE A ";P$;"  PIE.
8  ? "WHAT COLOR SHOULD I DRAW IT?
9  INPUT A$:I$=A$:D=10
10  GR. 3:POKE 752,1
11  IF I$=C$ THEN ? "YUMM!":GOTO 14
12  D=12:? "YEEEECH!
13  ? "WHO LIKES  ";A$;"  ";P$;"  PIES!"
14  SET. 0,H,L:COLOR 1
15  FOR X=1 TO 8
16  FOR Y=0 TO 360 STEP 20
17  PLOT Y*SIN(Y)+20,X*COS(Y)+8
18  SO. 0,Y,D,8:NEXT Y:NEXT X
19  SO. 0,0,0,0:NEXT N:GOTO 2
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up the TV volume.
3. Type RUN.
4. Type in the right color.

How It Looks



5. If you're right, you will see a spinach pie and hear nice music.



6. When you want to end, press **SYSTEM** **RESET** .
7. Save on disk or cassette as *Piemaker.*

Music

Musical Keys

Now's the time to make music and write your own tunes. How does your name sound? How about other words? Which key has *no* sound at all? What happens when you hold the key down?

Keys Program

```
1  OPEN #1,4,0,"K:"
2  ? " ↰    PRESS ANY KEY FOR MUSIC
3  GET #1,I
4  S=I*10
5  SO. 1,S,10,10
6  SET. 2,I,4
7  FOR T=1 TO 120:NEXT T
8  SO. 1,0,0,0
9  GOTO 3
```

Variations

You can get really funny sounds by changing line 6.

Old line 6: SOUND 1,S,10,10

New line 6: SOUND 1,S,6,10

Test your friends. How many keys do they have to press before they find the one that makes *no* sound?

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Turn the TV volume up.
4. Press any key for music. The color and sound changes with each key pressed. How does *your* name sound?

How It Looks



5. Press **SYSTEM RESET** to end.
6. To begin again, type RUN.
7. Save on disk or cassette as *Keys*.

Chords

Anytime you press a key, a chord will play. Can your friends remember how each other's names sound? This is like musical keys, but instead of hearing one tone for each key, you'll hear a chord. Which key has a silent chord?

Chords Program

```
1  OPEN #1,4,0,"K:"
2  ? " ↰   PRESS ANY KEY FOR MUSIC
3  GET #1,S1
4  S1=10*S1
5  S2=3*S1
6  SO. 0,S1,10,8
7  SO. 2,S2,10,8
8  SO. 1,S1+2,10,6
9  SO. 3,S2+2,10,6
10  FOR T=1 TO 200:NEXT T
11  FOR C=0 TO 3
12  SOUND C,0,0,0
13  NEXT C
14  GOTO 3
```

Variations

You can change the sounds by changing line 5. For example:

Old line 5: S2=3*S1

New line 5: S2=8*S1

Or you can make buzzy sounding chords by changing lines 6 through 9 by changing each 10 to a 6. For example:

Old line 6: SOUND 0,S1,10,8

New line 6: SOUND 0,S1,6,8

Test your friends. How many keys do they have to press before they find the one that makes *no* sound?

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Turn up the TV volume.
4. Press any key for chords. Does your age sound good? The name of your school?

How It Looks



5. Press **SYSTEM RESET** to end.
6. To begin again, type RUN.
7. Save on disk or cassette as *Chords*.



Fog

It is a dark, dark night. There is a lot of fog. You are the captain of a boat trying to go to shore safely. When you toot your horn, you can tell whether you are near rocks or other boats. Then you can steer the boat to safety.

The computer will make two sounds. If the sounds are the same, you type "S" for *same*. If they are different, you type "D" for *different*. When you are right, the boat is safe. When you are wrong, the boat crashes. Be careful now!

Fog Program

```
1  DIM A$(1)
2  ? " ↰  ARE THE SOUNDS THE SAME OR DIFFERENT?
3  FOR T=1 TO 600:NEXT T
4  S=INT(RND(0)*100+100)
5  GR. 2:TRAP 5
6  S1=S
7  V=8:GOSUB 20
8  IF RND(0)>0.5 THEN S=INT(RND(0)*100+100)
9  V=0:GOSUB 20
10  V=8:GOSUB 20
11  ? "S=SAME  D=DIFFERENT
12  SO. 1,0,0,0
13  POS. 8,4
14  INPUT A$
15  IF S=S1 AND A$="S" OR S< >S1 AND A$="D" THEN 18
16  SET. 4,3,4
17  ? #6;"CRASH":GOTO 19
18  SET. 4,12,4:? #6;"SAFE
19  FOR T=1 TO 600:NEXT T:GOTO 4
20  SO. 1,S,10,V:FOR T=1 TO 300:NEXT T:RETURN
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up the TV volume.
3. Type RUN.
4. Listen to the two foghorn sounds.
5. Type "S" if they are the same. Type "D" if they are different.

How It Looks

```
S=SAME  D=DIFFERENT
? ■
```

6. Press **RETURN** .
7. If you are right the boat is "SAFE." If you are not, the screen says "CRASH." Then the next pair will be presented.

```
SAFE

S=SAME     D=DIFFERENT
? SAME
```

8. To end, press **SYSTEM RESET** .
9. Save on disk or cassette as *Fog*.



Hear Ye!

In this game you will hear pairs of sounds. Listen to the two sounds and look at the colors and numbers that show up on the screen. See if you can remember the pairs of sounds you heard in the right order.

Each sound is always with the same number and same color. The computer will ask you to type in the numbers of the sounds you heard. You should type them in one at a time. Wait till the sound stops before pressing the next number.

If you type in a wrong number, the next pair of sounds will appear. If you type in the two numbers in the right order, you will get a point. After you have five points, the game gets harder! You will hear the sound and see the color, but not the number. Now you will have to remember which numbers went with which sounds and colors and type them in from memory.

Hear Program

```
1  DIM T(99):OPEN #1,4,0,"K:":R=0
2  ? " ↰   LISTEN TO THE 2 TONES
3  ? "LOOK AT THE COLORS AND NUMBERS
4  FOR T=1 TO 1200:NEXT T
5  GR. 0:FOR C=1 TO 2
6  T(C)=INT(RND(0)*4+1):S=T(C)
7  GOSUB 17:NEXT C
8  ? " ↰   NOW TYPE IN THE NUMBERS
9  POKE 764,255
10  FOR C=1 TO 2:GET #1,I:S=I-48
11  GOSUB 17
12  IF S< >T(C) THEN?"↰NOPE! YOUR SCORE IS  ";R:SO. 1,60,8,10:FOR
    T=1 TO 100:NEXT T:SO. 1,0,0,0:GOTO 16
13  NEXT C
14  R=R+1
15  ? " ↰   GREAT! YOUR SCORE IS  ";R
16  FOR T=1 TO 700:POKE 764,255:NEXT T:GOTO 5
17  S1=200-4*S:SO. 1,S1,10,8:POS. 10,10:POKE 752,1
18  IF R<5 THEN ? S
19  SET. 2,S1,5:FOR T=1 TO 400:NEXT T
20  SO. 1,0,0,0:FOR T=1 TO 200:NEXT T:SET. 2,9,4:RETURN
```

Variations

To make it easier, delete line 18. The numbers will continue to appear after the score of 5 is reached.

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up the TV volume.
3. Type RUN.
4. Listen to the sound. Look at the color and number. Try to remember them. (*Hint:* Say the numbers out loud if the child finds this *difficult.*)

How It Looks



5. Type in each number, one at a time. Remember to wait till the sound stops before you press the next number. (If you type in the wrong number, the computer will say NOPE and go on to the next set.)





6. After your score is 5, the sound and color appear, but the number does not.



7. To begin again, press **SYSTEM RESET** .
8. Type RUN.
9. Save on disk or cassette as *Hear.*

Tally Sheet

What's your best score?

Name	*Day*	*Score*	*Name*	*Day*	*Score*

Listen

This is a harder version of *Hear Ye! Hear Ye!* In this game there is a sequence of four sounds. Listen to the four sounds and look at the colors and numbers that show up on the screen. See if you can remember the order of the sounds you heard. The game gets harder because after your score is 5, the numbers disappear.

Each time you hear a sound you will also see a number and a color. Each sound is always with the same number and color. The computer will ask you to type in the number of the sounds you heard. You should type them in one at a time and wait till the sound stops before pressing the next number. If you type in a wrong number, you will hear a whistle and the next set of four sounds will appear. If you type in the four numbers in the right order, you will get a point. After you have 5 points, the game gets harder! You will hear the sound and see the color, but not the number. Now you have to remember which numbers went with which sounds and colors and type them in from memory.

(*Note:* If you have already entered and saved *Hear Ye!* all you need to do is change lines 2, 5, and 10 to those shown below.)



Listen Program

```
1  DIM T(99):OPEN #1,4,0,"K:":R=0
2  ? " ↰   LISTEN TO THE 4 TONES
3  ? "LOOK AT THE COLORS AND NUMBERS
4  FOR T=1 TO 1200:NEXT T
5  GR. 0:FOR C=1 TO 4
6  T(C)=INT(RND(0)*4+1):S=T(C)
7  GOSUB 17:NEXT C
8  ? " ↰   NOW TYPE IN THE NUMBERS
9  POKE 764,255
10  FOR C=1 TO 4:GET #1,I:S=I-48
11  GOSUB 17
12  IF S< >T(C) THEN?"      NOPE! YOUR SCORE IS  ";R:SO. 1,60,8,10:FOR
    T=1 TO 100:NEXT T:SO. 1,0,0,0:GOTO 16
13  NEXT C
14  R=R+1
15  ? "      GREAT! YOUR SCORE IS  ";R
16  FOR T=1 TO 700:POKE 764,255:NEXT T:GOTO 5
17  S1=200-4*S:SO. 1,S1,10,8:POS. 10,10:POKE 752,1
18  IF R<5 THEN ? S
19  SET. 2,S1,5:FOR T=1 TO 400:NEXT T
20  SO. 1,0,0,0:FOR T=1 TO 200:NEXT T:SET. 2,9,4:RETURN
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up the TV volume.
3. Type RUN.

How It Looks



4. Listen to the sounds. Remember the colors and numbers. Type in each number, one at a time. Wait till the sound stops before you press the next number. If you type in the wrong number, the computer will say "Nope" and go on to the next set of sounds.



5. After your score is 5, the sound and color appear, but the number doesn't.



6. To begin again, press **SYSTEM RESET** .
7. Type RUN.
8. Save on disk or cassette as *Listen*.

Tally Sheet

What's your high score?

Name	*Day*	*Score*	*Name*	*Day*	*Score*

High-Low

Can you tell whether the musical tones are higher or lower than each other? One note will play. Then a second note will play. Is the second note lower or higher? There are five comparisons in a round. Each time you are right, the game gets harder, because the tones get closer together. How many of the five can you get right? Will your score be high or low?

Highlow Program

```
1  DIM A$(1):S=0
2  N1=150:D=80
3  FOR C=1 TO 5
4  R=INT(2*RND(0))
5  POKE 752,1
6  ? " ↰   LISTEN TO THE TONES
7  IF R=0 THEN N2=N1-D:GOTO 9
8  N2=N1+D
9  N=N1:GOSUB 20:N=0:GOSUB 20
10  N=N2:GOSUB 20:N=0:GOSUB 20
11  ? "WAS THE SECOND TONE
12  ? "    H=HIGHER  L=LOWER
13  POKE 764,255:INPUT A$
14  IF A$="H" AND R=0 THEN 17
15  IF A$="L" AND R=1 THEN 17
16  ? "TRY ANOTHER ONE":GOTO 18
17  ? "GREAT!! IT'S GONNA GET HARDER!":S=S+1:D=D/2
18  FOR T=1 TO 900:NEXT T:NEXT C
19  ? "SCORE. . .";S:? "ROUND OVER":END
20  SO. 0,N,10,8:FOR T=1 TO 300:NEXT T:RETURN
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up the TV volume.
3. Type RUN.
4. Listen to the first tone. Then listen to the second.

How It Looks



5. Is the second tone higher *or* lower than the first? Type in the first letter of the correct response (H or L).



6. The computer will tell you if you're right or wrong.
7. Press **RETURN** .



8. After five tries, the round ends. Are you a Beethoven?



9. Type RUN to play again.

10. Save on disk or cassette as *Highlow.*

Tally Sheet

What's your high score?

Name	*Day*	*Score*	*Name*	*Day*	*Score*

Sound Off

How many sounds can you remember?

Each round of *Sound Off* gets longer. First you hear two sounds and see a number and a color. You type in the numbers you saw. When you get that right, you'll hear three sounds. Each time you type in the right numbers, the next round will get longer. If you are wrong, the next round will not get longer.

Soundoff Program

```
1  DIM T(99):OPEN #1,4,0,"K:":R=1
2  ? " ↰   LISTEN TO THE TONES
3  ? "LOOK AT THE COLORS AND NUMBERS
4  FOR T=1 TO 1200:NEXT T
5  GR. 0:FOR C=1 TO R+1
6  T(C)=INT(RND(0)*4+1):S=T(C)
7  GOSUB 17:NEXT C
8  ? " ↰   NOW TYPE IN THE NUMBERS
9  POKE 764,255
10  FOR C=1 TO R+1:GET #1,I:S=I-48
11  GOSUB 17
12  IF S< >T(C) THEN ?" ↰   NOPE! TRY ANOTHER ONE":SO. 1,60,8,10:
    FOR T=1 TO 100:NEXT T:SO. 1,0,0,0:GOTO 16
13  NEXT C
14  R=R+1
15  ? "YOU REMEMBERED  ";R;" SOUNDS!
16  FOR T=1 TO 700:POKE 764,255:NEXT T:GOTO 5
17  S1=200-4*S:SO. 1,S1,10,8:POS. 10,10:POKE 752,1:? S
18  SET. 2,S1,5
19  FOR T=1 TO 400:NEXT T
20  SO. 1,0,0,0:FOR T=1 TO 200:NEXT T:SET. 2,9,4:RETURN
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Turn up the TV volume.
3. Type RUN.
4. Listen to the sounds. Remember the colors and numbers. Type in each number, one at a time. Wait till the sound stops before you press the next number. If you type in the wrong number, the computer will say "Nope" and go on to the next set of sounds.
5. Type in the numbers one at a time. Wait till the sound stops before typing the next number.

How It Looks



6. If you were right, you will see:

If you type in the wrong number, the computer will say NOPE! Then the next set will be presented.

```
NOW TYPE IN THE NUMBERS
1
YOU REMEMBERED 2 SOUNDS!
```

7. To start over, press **SYSTEM RESET** and type RUN.

8. Save on disk or cassette as *Soundoff.*



Tally Sheet

What's your best score?

Name	*Day*	*Score*	*Name*	*Day*	*Score*

Free Play

Graphics

Graphics is a terrific program for little children who are still in the "hitting every key" phase. Everytime you hit a key a colorful graphics character comes up. Holding a key down repeats the pattern.

For older kids: you can change colors and also print messages by using certain keys.

Graphics Program

```
1  GR. 18
2  POKE 756,226
3  OPEN #1,4,0,"K:"
4  GET #1,I
5  L=ABS(I-64)
6  ? #6;CHR$(L);
7  IF PEEK(84)=12 THEN POS. 0,0
8  GOTO 4
```

Options

Hitting any key gives graphics. But some keys are special!

1. Holding the key down repeats the pattern.
2. Press the  key once and you can change color to purple. Just press the key again to go to orange.
3. Press the  key once. Then press the  key once. This changes the color to blue. Just hit the key again to return to orange.
4. Press the  key once. Then hold the  key down. You can write in lower case now. Just hit the key again to return to the original graphics.
5. Explore! Have fun! The Atari has a big bag of tricks!
6. Save on disk or cassette as *Graphics*.

Up-Down! Left-Right!

This program is a free-play exercise for children learning to use the joystick and learning the meaning of *up, down, left* and *right.* When the child moves the joystick, an orange square appears.

Someone older than the child should point to the top, bottom or sides of the screen. The child should move the joystick up, down, left or right and the orange square will appear. *Can the* child move the joystick in the right direction for the square to appear?

If the child can do the above activity correctly, then he can be asked with words instead of pointing, "Move *up, down, left* or *right.*"

Up Program

```
1  GR. 21
2  S=STICK(0)
3  IF S=14 THEN H=40:V=10
4  IF S=13 THEN H=40:V=45
5  IF S=11 THEN H=15:V=30
6  IF S=7 THEN H=65:V=30
7  IF S=15 THEN 1
8  PLOT H,V
9  DRAWTO H,V-10
10  DRAWTO H-10,V-10
11  DRAWTO H-10,V
12  DRAWTO H,V
13  FOR T=1 TO 500
14  NEXT T
15  GOTO 1
```



How It Works

1. Type in the program or load it from disk or cassette.
2. Plug the joystick into jack 1.
3. Type RUN.
4. Show the child how to hold the joystick so that the red button is on the upper left and "TOP" is pointing toward the TV screen.
5. Move the joystick *up* and the screen will look like this . . .

How It Looks



6. Move *down*. If the child is too young to know verbal directions, point to where on the screen the square should be.



7. Move *left*.



8. Move *right*.



9. To end, press **SYSTEM RESET** .

10. Save on disk or cassette as *Up*.

Trigger Writing

Little kids can spurt letters, numbers and graphics on the screen using the red button on the joystick. You can type in BIG LETTERS or designs.

To use the special graphics characters, hold down the CTRL key while you push any letter key when you are asked which letters or graphics you want to use. You can also use the  key to get white on blue designs.

When you have the combination you want, just press RETURN . You can move the joystick to the part of the screen you want. When you want to print, press the red button on the joystick.

Trigger Program

```
1  DIM A$(30)
2  ? " ↰  TYPE LETTERS OR GRAPHICS
3  ? "MOVE STICK. PRESS RED BUTTON.
4  INPUT A$
5  ? "  ↰  "
6  POS. 20,10
7  ? "  ";
8  J=STICK(0)
9  IF J=11 THEN ? "←";
10  IF J=7 THEN ? "→";
11  IF J=14 THEN ? "↑";
12  IF J=13 THEN ? "↓";
13  IF STRIG(0)=0 THEN ? A$;
14  GOTO 8
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Type in the combination of letters, numbers or symbols that you want on one line only.

How It Looks



4. This person typed T T. Then she pressed the key so everything else would be blue on white. Then she held down the CTRL key and typed P, M, . The result of this was a graphics design like you see on the right.



5. Press RETURN .
6. Move the joystick up and down and side to side and press the red button to draw a design.
7. Press SYSTEM RESET to end.
8. To begin again, type RUN.
9. Save on disk or cassette as *Trigger*.



Joystick Drawing

Now you can draw designs using the joystick. You can type in anything you want: BIG LETTERS or small letters, or even the words "Big" and "Small." If you want, you can use numbers, too: 123456789 or 777440044777.

When you have the combination you want, just press **RETURN** and move the joystick in any direction to draw all kinds of things. Remember, you can use more than one letter, number or symbol at a time, but it all has to be on one line only.

To use symbols: Hold down the **CTRL** key while you press any letter keys. If you press the  key first, your design will be all white on blue.

Draw Program

```
1  DIM A$(30)
2  ? " ↰  TYPE LETTERS OR GRAPHICS
3  ? "PRESS RETURN
4  ? "MOVE STICK TO DRAW
5  INPUT A$
6  ? "  ↰  "
7  POS. 20,10
8  ? ".";
9  J=STICK(0)
10  IF J=11 THEN ? A$;
11  IF J=7 THEN ? "→";A$;"→";
12  IF J=14 THEN ? "↑";A$;"↑";
13  IF J=13 THEN ? "↓";A$;"↓";
14  GOTO 9
```

How It Works

1. Type in the program or load it from disk or cassette.
2. Type RUN.
3. Type in the combination of letters, numbers or symbols that you want on one line only.

How It Looks



4. This person typed T T. Then he pressed the key so everything else would be blue on white. Then he held down the CTRL key and typed P, M, . The result of this was a graphics design like you see on the right.
5. Press RETURN .



6. Move the joystick up and down and side to side to draw a design.



7. Press SYSTEM RESET to end.
8. To begin again, type RUN.
9. Save on disk or cassette as *Draw*.

Paddle Play

You can get music and dancing dots using your paddle controllers! When you turn the paddle, the dots move, notes play, and the color changes. When you press the red button, the sounds change.

Paddle Program

```
1  TRAP 10
2  ? "TURN THE PADDLE AND PRESS RED BUTTON
3  FOR T=1 TO 900:NEXT T
4  GR. 0
5  SET. 2,P,2
6  COLOR 1
7  LET P=PADDLE(0)/6.2
8  IF P<38 THEN POKE 82,P
9  ?" . "
10  SO. 0,P,10,8
11  IF PTRIG(0)=0 THEN SO. 0,P,6,P
12  GOTO 5
```



Variations

To change the dots to some other design, change line 11. You can press any key instead of T.

Old line 11: ? "●"

New line 11: ? "◢" Note: Press ESC and then hold down CTRL and press H key. Repeat with as many keys as you like.

How It Works

1. Type in the program or load it from disk or cassette.
2. Plug paddle into jack 1.
3. Turn up the TV volume.
4. Type RUN.
5. Move paddle from side to side. Press the red button occasionally.

How It Looks



6. Your screen will look something like this.



7. To end, press **SYSTEM RESET** .
8. Save on disk or cassette as *Paddle*.

$12.95

Rainy Day Activities for the Atari®

Nancy Mayer

A toy store in your family living room? Fifty games and activities ranging from music and art to letters, numbers, and nonsense readily available for your young children? That's exactly what you'll have with this exciting new book and your ATARI Home Computer. Nancy Mayer has put together one of the cleverest books on the market today. Aiming at the 3-9 year olds (who are usually neglected in today's computer market), Mayer has developed a charming book for parents to share with their youngsters. No programming is involved! All you do is type in a short program listing (none are longer than 20 lines), and let the fun begin.

A few of the games require no more than the ability to hold and move a joystick; some require reading skills or an adult or older child nearby to help. The skill level for many activities is suggested at the beginning of the game description, and best of all, your kids will be learning as they play. You won't even have to wait for a rainy day to involve the whole family in computer learning and entertainment.

Some of the games in this book are:

Popcorn • Lasso the Stars • Pigtalk • Soundoff • Count the Sides • Graffiti • Snowfall Ending • Musical Keys

cover illustration by Donna Ward

RESTON PUBLISHING COMPANY, INC.
A Prentice-Hall Company
Reston, Virginia 22090

0-8359-6390-X

0 21898 63900 4